Related Rates Problems

Objective This lab assignment provides additional practice with related rates problems.

Background Related rates problems are one of the principle applications of the Chain Rule. (The other principle application of the Chain Rule is implicit differentiation.) The key to solving a related rates problem is the identification of appropriate relationships between the variables in the problem — and putting all of the pieces of information together to produce an answer to the question.

General Strategy for Solving Related Rates Problems

Step 1: Read the entire problem; identify quantities to be found.
Step 2: Draw a diagram; include relevant labels.
Step 3: Identify constants, values of functions, rates of change, and quantities that are functions of time.
Step 4: Write equation(s) relating quantities in the problem.
Step 5: Use the Chain Rule to differentiate both sides of an equation with respect to time.
Step 6: Substitute given constants, values of functions, and rates of change.
Step 7: Solve for the quantities identified in Step 1.

Look for these steps in the Examples below, then duplicate this approach to answer the Questions.

Some of the steps are easier to do with pencil and paper than with Maple. Step 1–3 are often easier to complete manually. Step 5 is one step that Maple is most likely to be useful. In general, the command `diff( eq, t );` differentiates both sides of the equation `eq` with respect to time `t`. Remember that all functions of time must show this dependence in `eq`.

Discussion Enter, and execute, the following Maple commands in a Maple worksheet.

Example 1: Melting Snowball

PROBLEM STATEMENT:
If a snowball melts so that its surface area decreases at a rate of 1 cm^2/min, find the rates of change of the radius and volume when the diameter is 10 cm.

SOLUTION:
Steps 1 and 2 are not explicitly included; draw the picture for yourself. Time-dependent functions are the radius, $r(t)$, the surface area, $S(t)$, and the volume, $V(t)$. (Recall that, for a sphere with radius r, $V = \frac{4}{3}\pi r^3$ and $S = 4\pi r^2$.)

Step 3:
```
> Rvalue := r(t) = 5;                                  # given value
> Srate := diff( S(t), t ) = -1;                       # given rate
```
Step 4:
```
> Seq := S(t) = 4*Pi*r(t)^2;                           # relate S, r
> Veq := V(t) = 4/3*Pi*r(t)^3;                         # relate V, r
```
Step 5:
```
> dSeq := diff( Seq, t );                              # Chain Rule
> dVeq := diff( Veq, t );                              # Chain Rule
```
Step 6:
```
> q1 := eval( Veq, Rvalue );                           # V when r = 5
> q2 := isolate( dSeq, diff( r(t), t ) );              # solve for dr/dt
> q3 := lhs(q2)                                        # dr/dt when
>       = eval( rhs(q2), { Rvalue, Srate } );          # dS/dt = -1, r = 5
```
Step 7:
```
> eval( dVeq, { Rvalue, q3 } );                        # dV/dt when r = 5
> eval( Veq, Rvalue );                                 # V when r = 5
```

Example 2: Colliding Cars

PROBLEM STATEMENT:
Car 1 is traveling east at v_1 km/hr and Car 2 is traveling south at 50 km/hr. How fast is Car 1 traveling if the rate at which the two cars approaching each other is 80 km/hr at the instant when Car 1 is 3.5 km and Car 2 is 5.1 km from the intersection?

SOLUTION:
Let $x(t)$ and $y(t)$ denote distances of Car 1 and Car 2, respectively, from the intersection. Let $d(t)$ be the distance between the cars.

Step 3:
```
> dist1 := x(t) = 3.5;                      # given value
> dist2 := y(t) = 5.1;                      # given value
> vel1 := diff( x(t), t ) = v1;             # given rate
> vel2 := diff( y(t), t ) = -50;            # given rate
> vel3 := diff( d(t), t ) = -80;            # given rate
```

Step 4:
```
> eq := d(t)^2 = x(t)^2 + y(t)^2;           # distance (squared)
```

Step 5:
```
> Deq := diff( eq, t );                     # Chain Rule
```

Step 6:
```
> q1 := eval( eq, { dist1, dist2 } );       # distance (squared)
> q2 := d(t) = sqrt( 38.26 );               # easier than solve
> q3 := eval( Deq, { vel1, vel2, vel3 } );  # insert rates
> q4 := eval( q3, { q2, dist1, dist2 } );   # insert values
```

Step 7:
```
> q5 := solve( q4, { v1 } );                # solve for v1
```

Note

(1) In the process of solving Example 2 you might have wondered "Why didn't we work with the explicit formula for the distance?" Let's look at how this would go. After taking the square root of both sides of the distance formula: $d(t) = \sqrt{x(t)^2 + y(t)^2}$ the differentiation in Step 5 is somewhat more complicated. The result is:

$$\frac{d}{dt}d(t) = \frac{x(t)\frac{dx}{dt} + y(t)\frac{dy}{dt}}{\sqrt{x(t)^2 + y(t)^2}}.$$

Note that this is simply the equation `Deq` divided by $d(t) = \sqrt{x(t)^2 + y(t)^2}$. From here the remainder of the problem is identical. Which method you prefer is a personal choice.

Questions

(1) A camera tracks the launch of a spacecraft during a perfectly vertical launch. The camera is located on the ground 2 miles from the launch pad. If the rocket is 2.5 miles above ground and traveling at 700 mi/hr, at what rate is the camera angle (measured from the horizontal) changing? Give your answer in radians/hr, radians/s, and degrees/s.

(2) A conical ice cream cone has a vertical axis, is 10 centimeters high, and has an opening with diameter 6 centimeters. The cone is completely filled with your favorite flavor of ice cream, but you do *not* eat the ice cream or the cone. Two hours later, a hole develops in the bottom of the cone and the melted ice cream drains at a rate of 2 cubic centimeters per second.
   (a) Find the relationship between the rates of change of the height and radius of the ice cream remaining in the cone.
   (b) What is the rate of change of the height of (melted) ice cream left in the cone when the height is 5 centimeters?